

# Pirate Dogfight

It's you against the Sky Pirate. He moves ahead, you accelerate - He drops behind, you slow down. You must try to get level with him and then you can fire, hoping that he won't be able to fire and hit you first.

Use the letter keys A to accelerate, D to decelerate and F to fire. Your computer will tell you your speed and position relative to the pirate. You will need to be ready to press the appropriate keys as soon as you press RUN. Keep pressing A and D until you get level and then fire.



```
▲●10 CLS
20 PRINT "PIRATE DOGFIGHT"
```

```
★▲●30 LET V=INT(RND*11-5)
```

```
★▲●40 LET S=-INT(RND*3+1)
```

```
50 IF ABS(S)>20 THEN GOTO 230
▲●60 CLS
70 PRINT "YOU ARE ";
80 IF S<0 THEN PRINT "BEHIND"
90 IF S>0 THEN PRINT "AHEAD"
100 IF S=0 THEN PRINT "LEVEL"
110 PRINT "YOU ARE GOING ";
120 IF V>0 THEN PRINT "FASTER"
130 IF V<0 THEN PRINT "SLOWER"
140 IF V=0 THEN PRINT "SAME"
```

```
★▲●150 LET I$=INKEY$
160 IF I$="A" THEN LET V=V+1
170 IF I$="D" THEN LET V=V-1
180 IF I$="F" AND S=0 THEN GOTO 250
```

## How the program works

Chooses a number between -5 and +5 for your speed relative to pirate and puts it in V.

Chooses a number for your distance from the pirate and puts it in S. This is negative at the start, which means you are behind him.

Checks if the distance between you is more than 20. If so, computer jumps to 230 to say you've lost sight of him.

Checks the values of S and V and prints your position and speed in relation to the pirate.

Looks to see if you are pressing a key and, if so, which one. If it is A, your speed is increased by 1. If D, your speed is decreased by 1. If you are pressing F and S=0 (i.e. you're level) then it jumps to 250. (If S is not 0 when you press F then nothing happens.)

```
190 LET S=S+V
★▲●200 FOR I=1 TO 20
210 NEXT I
220 GOTO 50
230 PRINT "YOU LOST SIGHT OF HIM"
240 GOTO 330


250 IF ABS(V)<2 THEN GOTO 290

260 PRINT "YOU ONLY MANAGED TO"
270 PRINT "SCARE HIM."
280 GOTO 40

★▲●290 IF RND>.7 THEN GOTO 320
300 PRINT "YOU SHOT HIM DOWN"
310 GOTO 330
320 PRINT "HE SHOT FIRST"
330 STOP
```

The above listing will work on a ZX81. For other computers, make the changes below.

```
●10,60 HOME
▲10,60 PRINT CHR$(147)
★▲●30 LET V=INT(RND(11)*11-5)
●30 LET V=INT(RND(0)*11-5)
★▲●40 LET S=-INT(RND(1)*3+1)
●40 LET S=-INT(RND(0)*3+1)
●145 I$=""
★150 I$=INKEY$(1)
●150 IF PEEK(-16384)>127 THEN GET I$
▲150 GET I$
★▲●200 FOR I=1 TO 200
●290 IF RND(0)>.7 THEN GOTO 320
★▲●290 IF RND(1)>.7 THEN GOTO 320
```



## How to make the game easier

You may find this game quite difficult. To make it slightly easier, add these two lines. You will then be able to see the relative positions of the two planes printed on the screen.

```
195 IF ABS(S)>10 THEN GOTO 200
196 PRINT TAB(W/2); "HIM"
197 PRINT TAB(S+W/2); "YOU"
Replace w with the width of your screen.
```

## Puzzle corner



The random test in line 290 is loaded in your favour. How could you change it so your chances of winning are equal?